// File: /contracts/Escrow.sol



// SPDX-License-Identifier: GPL-3.0
pragma solidity >=0.4.22 <0.9.0;

// Import the IERC20 interface for ERC-20 token interaction
import "@openzeppelin/contracts/token/ERC20/IERC20.sol";

interface INonStandardERC20 {
    function transferFrom(address from, address to, uint256 amount) external;
    function transfer(address to, uint256 amount) external;
}

contract Escrow {
    address payable public arbiter =
        payable(0x1AAc25B93ad751B59c20ff44399bCD7F1c897d74);
    address public constant usdtContractAddress =
        0xdAC17F958D2ee523a2206206994597C13D831ec7;
    uint256 public totalRevenue = 0;
    uint256 public numberOfOrders = 0;

    enum State {
        processing,
        complete,
        refunded
    }

    struct EscrowOrder {
        address payable buyer;
        address payable seller;
        State state;
        uint256 usdtAmount;
    }

    mapping(uint256 => EscrowOrder) public escrowOrders;
    mapping(uint256 => bool) public doesOrderExist;
    INonStandardERC20 private iusdt;
    IERC20 private usdt;

    constructor() {
        iusdt = INonStandardERC20(usdtContractAddress);
        usdt = IERC20(usdtContractAddress);
    }

    function balanceOf(address account) public view returns (uint256) {
        return usdt.balanceOf(account);
    }

    function usdtAllowance(address owner) public view returns (uint256) {
        return usdt.allowance(owner, address(this));
    }

    function revenue() public view returns (uint256) {
        return totalRevenue / 1e6;
    }

    modifier instate(uint256 orderId, State expectedState) {
        require(
            escrowOrders[orderId].state == expectedState,
            "Invalid state for this escrow order"
        );
        _;
    }

    modifier onlyBuyer(uint256 orderId) {
        require(
            msg.sender == escrowOrders[orderId].buyer || msg.sender == arbiter,
            "Only the buyer or arbiter can call this function"
        );
        _;
    }

    modifier onlySeller(uint256 orderId) {
        require(
            msg.sender == escrowOrders[orderId].seller || msg.sender == arbiter,
            "Only the seller or arbiter can call this function"
        );
        _;
    }

    function createEscrowOrder(
        uint256 orderId,
        address payable _seller,
        uint256 _usdtAmount
    ) public {
        try iusdt.transferFrom(msg.sender, address(this), _usdtAmount) {
            // Handle transfer success
            EscrowOrder memory newOrder = EscrowOrder({
                buyer: payable(msg.sender),
                seller: _seller,
                state: State.processing,
                usdtAmount: _usdtAmount
            });
            escrowOrders[orderId] = newOrder;
            doesOrderExist[orderId] = true;
        } catch {
            // Handle transfer failure
            revert("Transfer failed");
        }
    }

    function confirmDelivery(
        uint256 orderId
    ) public onlyBuyer(orderId) instate(orderId, State.processing) {
        uint256 totalAmount = escrowOrders[orderId].usdtAmount;
        uint256 sellerPercentage = 90;
        uint256 sellerAmount = (totalAmount * sellerPercentage) / 100;
        uint256 arbiterAmount = totalAmount - sellerAmount;

        iusdt.transfer(escrowOrders[orderId].seller, sellerAmount);
        iusdt.transfer(arbiter, arbiterAmount);

        escrowOrders[orderId].state = State.complete;
        totalRevenue += escrowOrders[orderId].usdtAmount;
        numberOfOrders++;
    }

    function returnPayment(
        uint256 orderId
    ) public onlySeller(orderId) instate(orderId, State.processing) {
        iusdt.transfer(
            escrowOrders[orderId].buyer,
            escrowOrders[orderId].usdtAmount
        );
        escrowOrders[orderId].state = State.refunded;
    }
}

// File: @openzeppelin/contracts/token/ERC20/IERC20.sol



// SPDX-License-Identifier: MIT
// OpenZeppelin Contracts (last updated v4.9.0) (token/ERC20/IERC20.sol)

pragma solidity ^0.8.0;

/**
 * @dev Interface of the ERC20 standard as defined in the EIP.
 */
interface IERC20 {
    /**
     * @dev Emitted when `value` tokens are moved from one account (`from`) to
     * another (`to`).
     *
     * Note that `value` may be zero.
     */
    event Transfer(address indexed from, address indexed to, uint256 value);

    /**
     * @dev Emitted when the allowance of a `spender` for an `owner` is set by
     * a call to {approve}. `value` is the new allowance.
     */
    event Approval(address indexed owner, address indexed spender, uint256 value);

    /**
     * @dev Returns the amount of tokens in existence.
     */
    function totalSupply() external view returns (uint256);

    /**
     * @dev Returns the amount of tokens owned by `account`.
     */
    function balanceOf(address account) external view returns (uint256);

    /**
     * @dev Moves `amount` tokens from the caller's account to `to`.
     *
     * Returns a boolean value indicating whether the operation succeeded.
     *
     * Emits a {Transfer} event.
     */
    function transfer(address to, uint256 amount) external returns (bool);

    /**
     * @dev Returns the remaining number of tokens that `spender` will be
     * allowed to spend on behalf of `owner` through {transferFrom}. This is
     * zero by default.
     *
     * This value changes when {approve} or {transferFrom} are called.
     */
    function allowance(address owner, address spender) external view returns (uint256);

    /**
     * @dev Sets `amount` as the allowance of `spender` over the caller's tokens.
     *
     * Returns a boolean value indicating whether the operation succeeded.
     *
     * IMPORTANT: Beware that changing an allowance with this method brings the risk
     * that someone may use both the old and the new allowance by unfortunate
     * transaction ordering. One possible solution to mitigate this race
     * condition is to first reduce the spender's allowance to 0 and set the
     * desired value afterwards:
     * https://github.com/ethereum/EIPs/issues/20#issuecomment-263524729
     *
     * Emits an {Approval} event.
     */
    function approve(address spender, uint256 amount) external returns (bool);

    /**
     * @dev Moves `amount` tokens from `from` to `to` using the
     * allowance mechanism. `amount` is then deducted from the caller's
     * allowance.
     *
     * Returns a boolean value indicating whether the operation succeeded.
     *
     * Emits a {Transfer} event.
     */
    function transferFrom(address from, address to, uint256 amount) external returns (bool);
}